# AN1292

# Adding a Voice User Interface to M68HC05 Applications

By Derrick B. Forte and Hai T. Nguyen CSIC Development Tools Austin, Texas

#### Introduction


As embedded microcontroller-based products become more sophisticated, additional emphasis is being placed on the design and implementation of their user interfaces. Visually based interfaces are commonly implemented with LCDs, LEDs, fluorescent displays, and lights. Many of these components can be controlled directly by an application's processor without using additional components. Voicebased user interfaces, on the other hand, are often implemented with speech synthesizers, speech processors, sound generators, and digital signal processors which operate in conjunction with an application's main processor. In addition to a processor dedicated to the generation of speech, designs frequently require a memory device to hold the data used by the processor, a loudspeaker, and audio amplification circuitry. The added cost for components and space has limited the implementation of speech-based user interfaces to higher end products and products for the visually impaired. This application note discusses adding a voice-based user interface to an application based on the Motorola MC68HC(7)05J1A microcontroller. In particular, interfacing members of the M68HC05 MCU Family to the Information Storage Devices (ISD) 1000 and 2500 family of voice record/playback devices is highlighted. The development of an audible thermometer application concludes the discussion.



### **Design Alternatives**

Most applications that use speech as part or all of their user interface utilize any one of a number of speech processors and synthesizers on the market. The speech output by these devices usually is stored as noncompressed or compressed digital data in on-chip or external memory. In devices designed specifically for high-volume applications, speech data usually is stored in on-chip ROM.

However, most speech processors and synthesizers designed for the general market require more flexibility than can be offered by using onchip ROM. Data for these processors is provided by a programmed external memory device. Depending on whether audio amplification is done by the speech processor or not, external circuitry may be needed to interface the processor to a loudspeaker. A block diagram of a typical speech system is illustrated in **Figure 1**.



#### Figure 1. Typical Speech System Design

The ISD 1000 and 2500 series of voice record/playback devices discussed in this application note integrate all the circuitry needed to record and play back audio signals on a single device. The speech processor, memory, and audio amplifier functional blocks needed to implement a speech interface are all integrated on this device. The members of the device family differ in the length of their recording times, ranging from 10 seconds to 90 seconds. The devices are designed to operate in a number of standalone recording and playback modes or under the control of an external microcontroller. The ease with which these devices can be interfaced with microcontrollers makes them ideal

for adding a voice-based user interface to an application based on a member of Motorola's M68HC05 Family of MCUs. The remainder of this note discusses using one of these devices to add a voice-based user interface to a simple M68HC05 MCU-based application, namely a digital thermometer.

### **Audible Thermometer Feature Definition**

The system design of the audible thermometer begins with the definition of the application's feature set. The audible thermometer senses ambient temperature and outputs the temperature reading in a prerecorded human voice. The thermometer is capable of sensing temperatures from –55 to +125 degrees Celsius in 0.5-degree increments. The thermometer powers up and remains in a low-power idle state until the user presses a button. Pushing the button wakes up the thermometer, causing it to acquire and output a temperature reading. After completing these tasks, the system returns to a low-power idle state.

#### Audible Thermometer Hardware Design

The system design of digital thermometers is a well-established paradigm in the design of embedded systems applications. The audible thermometer follows this model and its hardware design can be divided into two main functional blocks:

- 1. Temperature acquisition and conversion Senses ambient temperature and converts the reading to the digital domain
- 2. Audio processing and output Outputs the temperature reading in a human voice

To illustrate the ease with which a voice interface can be added to a Motorola M68HC05 MCU-based application, the Motorola MC68HC(7)05J1A microcontroller was chosen as the main system processor for this application. This device is the simplest and the most inexpensive member of Motorola's M68HC05 Family of microcontrollers. The MC68HC(7)05J1A's main on-chip peripherals include an 8-bit freerunning timer and 14 bidirectional I/O pins. The MC68HC(7)05J1A's simplicity constrains the role that it plays in the hardware implementation of these two blocks.

The temperature acquisition and conversion block consists of circuitry that senses the application's ambient temperature and converts it to a suitable electrical signal for processing by the system's microcontroller. This block typically consists of a temperature sensor, signal conditioning circuitry, and an A/D converter. The temperature sensor is capable of varying a voltage or current signal in proportion to its ambient temperature. The signal is then processed by some form of analog conditioning circuitry. The conditioning circuitry design is heavily dependent on the accuracy, sensitivity, and noise rejection parameters of the application's specifications and its components. This circuitry may amplify, filter, and linearize the signal in preparation for its conversion to the digital domain by the A/D converter. Once in digital form, the signal can be processed by the microcontroller. In most M68HC05-based applications, the temperature sensor and conditioning circuitry generate and process an analog signal for use by the MCU's on-chip A/D converter peripheral. However, since the MC68HC(7)05J1A does not

have an on-chip A/D converter, an external A/D converter is needed to implement this block completely. The added cost and space required by an external A/D converter led to the selection of the Dallas Semiconductor DS1820 One-Wire Digital Thermometer to implement the temperature acquisition and conversion block in this application. The DS1820 is a 3-pin device that integrates the temperature sensor, conditioning circuitry, and A/D converter needed to implement this block on a single device. In addition, the DS1820 also has nine bytes of scratchpad RAM and two bytes of EEPROM memory. Using this device results in substantial cost and space savings. The temperature sensed by this device is available to the microcontroller as a 9-bit binary number which can be read serially from a single pin.

The audio processing and output block in this application serves to output the temperature read in a human voice. As mentioned earlier, the ISD 1000 and 2500 series voice record/playback devices contain most of the circuitry needed to implement this block. The high degree of integration provided by this device allows this block to be implemented using this device, a few passive components, and a loudspeaker. The device selected for use in the audible thermometer is the ISD2560. This device is capable of recording and playing back 60 seconds of sound and/or speech. The ISD2560 records by sampling a speech or sound signal at 8 kHz and storing the samples as discrete analog levels in storage cells. The ISD2560 has 480 K of such cells mapped in a memory space that is divided into 600 addresses. Sound recording can be initiated at any one of the 600 addresses and is stopped either by the manipulation of device control signals or by reaching the end of the device's memory space. To separate recordings, special end of message (EOM) markers are placed in memory at the end of each recording. This gives the ISD2560 the ability to record a number of separate recordings or messages and play them back as many times or in any sequence desired. The audible thermometer uses this feature of the ISD2560 to output a sequence of pre-recorded phrases that correspond to the temperature read by the DS1820. In the thermometer, the ISD2560 is pre-recorded with phrases for the numbers 0 through 19, the numbers 20 through 90 in increments of 10, and the words "one hundred," "point," "degrees," "negative," and "Celsius." (See the Design Manual for ISD1000A Family for recording instructions.) These phrases

#### Audible Thermometer Hardware Design

are recorded at addresses in the ISD2560's memory space that are 16 units apart starting at address \$0000. This allots a time of 1.5 seconds per phrase. The ISD2560 signals encountering an EOM marker by pulsing the /EOM pin low and then high. The signal can be used by an external controller to concatenate a sequence of messages.

Although the ISD2560 is capable of operating in a number of standalone or operational modes, the MC68HC(7)05J1A interfaces with the device at its microcontroller interface.

The following describes the ISD2560's microcontroller interface pins and their functions:

- A0–A9 Address lines 0–9: Inputs used to access the 600 addresses within the device's memory space. Although the number of lines allows the selection of 1024 addresses, only addresses 00 to 257 hex are valid.
- 2. /CE Chip Enable: An active low pin that enables recording and playback operations
- PD Powerdown: An active high pin that puts the device in a lowpower idle state.
- 4. P/R Playback/ Record: A pin that enables device recording when it is high and enables playback operations when it is low.
- 5. /EOM End of Message: An active low pin that pulses for 12.5 msec after the end of a message.
- /OVF Overflow: An active low pin that signals the end of the device's memory space. This signal can be used to cascade more than one ISD device together for greater message storage capacity.

After defining the system's hardware functional blocks of the audible thermometer and selecting the components that comprise the blocks, the system block diagram in **Figure 2** was derived for the audible thermometer.

Schematics for the application's hardware design are located in **Audible Thermometer Schematics**.



### Figure 2. Audible Thermometer System Block Diagram

#### Audible Thermometer Software Design

The audible thermometer's system software can be divided into the main program functions and the low-level functions that interface the MC68HC(7)05J1A to the DS1820 and the ISD2560. The low-level driver routines are discussed first, since the main program routines are built on them.

When given the proper command sequence, the Dallas Semiconductor DS1820 One-Wire Digital Thermometer is designed to acquire a temperature measurement within one second and convert it to a 9-bit digital word. The temperature measured is mapped into a range of 9-bit words that span from -55 to +125 degrees Celsius in 0.5-degree increments. The upper byte of a word indicates whether the temperature read is above or below 0 degrees Celsius. An upper byte value of \$FF corresponds to a negative temperature and a value of \$00 corresponds to a positive temperature. The lower byte values range from \$01 to \$FA for positive temperatures and from \$FF to \$92 for negative ones. When a temperature is read, the converted word is stored, least significant byte first, in the first two bytes of the DS1820's scratchpad RAM memory. The device interfaces with a microcontroller over a single serial line using a half-duplex serial protocol. The protocol prescribes that the MCU initiate and sustain all communications with the DS1820. This protocol supports a full-featured command set that provides the microcontroller with complete control over the DS1820's operation. The DS1820 command set includes commands to read and write scratchpad RAM memory, to read and write EEPROM memory, and to perform a temperature reading and conversion operation. Although the DS1820 is a multi-featured device, the audible thermometer only uses the commands required to perform a temperature reading and conversion operation and read the 9-bit data word from the DS1820. In this application, the DS1820 interfaces to the MC68HC(7)05J1A at its PB5 bidirectional input/output (I/O) pin. Since the DS1820's protocol is not a standard, the MC68HC(7)05J1A must manipulate or "bit bang" the PB5 pin to communicate with a DS1820.

The DS1820's serial protocol supports three communication functions: reset, read, and write.

A reset sequence initializes a DS1820 and prepares it to receive a command from the MCU. A DS1820 reset can be initiated only by the microcontroller and consists of a reset pulse from the microcontroller followed by an acknowledgment pulse from the DS1820. This requires that after driving the serial line to output the reset pulse, the MCU's I/O pin must be changed from an output to an input to receive the acknowledgment pulse. Since setting the I/O line as an input three-states the serial line, a pullup resistor is needed to pull the serial line high while the microcontroller is not driving it. If an acknowledgment pulse is not received from the DS1820 within 15 to 60 microseconds from the rising edge of the reset pulse, the DS1820 is considered to be inoperative. Figure 3 illustrates the timing requirements for a DS1820 reset operation.





Figure 3. DS1820 Reset Sequence

The MC68HC(7)05J1A sends commands and data to the DS1820 using the device's write protocol. The microcontroller initiates a write cycle or time slot by pulling the serial line low. A write cycle must be a minimum of 60 microseconds long with a minimum recovery time of 1 microsecond between cycles. Data is output least significant bit first with each bit requiring one complete write cycle. **Figure 4** illustrates the timing requirements for writing a 1 or 0 to the DS1820.


#### **Application Note**



Microcontroller writes a 1 bit to the DS1820.

#### Figure 4. Microcontroller to DS1820 Write Cycle

The MC68HC(7)05J1A reads data from the DS1820 using the device's read protocol. The microcontroller initiates a read cycle or time slot by pulling the serial line low for a minimum of one microsecond. The DS1820 outputs a valid bit 15 microseconds after the start of the read cycle. Therefore, the MCU must change the I/O line driving the serial line from an output to an input before the DS1820 starts to output data. The pullup resistor on the serial line pulls up the line until the DS1820 is ready to output a bit. A read cycle must be a minimum of 60 microseconds with minimum recovery time of 1 microsecond between cycles. The DS1820 outputs data least significant bit first with each bit requiring one full read cycle. **Figure 5** illustrates the timing requirements for reading a 1 or 0 from the DS1820.




Microcontroller reads a 1 bit from the DS1820.

#### Figure 5. DS1820 Read Cycle

The ISD2560 driver functions enable the device to play back a sequence of pre-recorded phrases under the direction of the MC68HC(7)05J1A. The MC68HC(7)05J1A performs this simple sequence of I/O port operations to cause the ISD2560 to output a single pre-recorded phrase:

- 1. Pulls the ISD2560's PD low, taking the device out of powerdown mode
- 2. Sets the ISD2560's P/R pin high, enabling playback operation
- 3. Places the starting address of the message on the ISD2560's address bus
- 4. Pulses the ISD2560's /CE pin low then high for a minimum of 100 nanoseconds
- 5. Waits for a falling edge on ISD2560's /EOM pin, indicating that an EOM marker has been encountered
- 6. Waits for the rising edge on the ISD2560's /EOM pin, indicating the end of the EOM pulse

Figure 6 illustrates a timing diagram for the ISD2560's signals.



#### Figure 6. ISD2560 Control Signals Timing Diagram

The audible thermometer's main program flow is:

- 1. Initialize the MC68HC(7)05J1A's I/O ports.
- 2. Put the MC68HC(7)05 into low-power stop mode.
- 3. Wait for the user to press the pushbutton.
- 4. Acquire a temperature reading from the DS1820.
- 5. Output the reading to the ISD2560.
- 6. Return to stop mode and wait for the user to press the pushbutton.

Consult **Main Program Flowchart** for a detailed flowchart of the main program's operation.

After initializing the MC68HC(7)05J1A's I/O ports, the MCU is placed in stop mode. Pressing the pushbutton generates an MCU IRQ interrupt that wakes the processor out of stop mode. The processor then uses low-level driver routines to start a DS1820 temperature acquisition and conversion operation and read a 9-bit data word from the DS1820. (Consult Appendix B for a flowchart of the temperature acquisition routine.) If an error occurs during the acquisition of the word, the thermometer is placed into stop mode. Otherwise, the MC68HC(7)05J1A processes the word and determines the sequence of phrases to be output by the ISD2560. The processor then finds the address of each phrase from a series of tables. The address of each phrase is placed in the proper order in a phrase buffer. (Consult Appendix C for the flowchart of the audio processing routine.) The MCU then uses the ISD2560 low-level routines to output the sequence of phrases whose addresses are in the phrase buffer. After outputting the phrase sequence, the MCU returns to stop mode.

### Summary

The ISD2560 1000 and 2500 series of voice record/playback devices permit the implementation of cost-effective, voice-based user interfaces in products based on Motorola's M68HC05 microcontrollers. The devices are designed with a microcontroller interface that easily interfaces with even the simplest member of the M68HC05 Family.


#### **Bibliography**

Motorola MC68HC705J1A Technical Data


ISD Information Storage Devices ISD2500 Series Preliminary Data Sheet

Design Manual for the ISD1000A Family


Dallas Semiconductor DS1820 One-Wire Digital Thermometer Data Sheet




### **Temperature Reading Procedure Flowchart**




### Audio Output Procedure Flowchart



### Audio Output Procedure Flowchart (Continued)



## Audible Thermometer Schematics



#### Source Code

#### Source Code

\* Motorola reseves the right to make changes without further notice \* to any product herein to improve reliability, function, or design. \* Motorola does not assume any liability arising out of the \* application or use of any product , circuit, or software described \* herein; neither does it convey any license under its patent rights \* nor the right of others. Motorola products are not designed, \* intended or authorised for use as components in systems intended \* for surgical implant into the body, or other applications intended \* to support life, or for any other application in which failure \* of the Motorola product could create a situation where personal \* injury or death may occur. Should Buyer purchase or use Motorola \* products for any such intended or unauthorised application, Buyer \* shall indemnify and hold Motorola and its officers, employees \* subsidiaries, affiliates, and distributors harmless against all \* claims, costs, damages, expenses and reasonable attorney fees \* arising out of, directly or indirectly, any claim of personal \* injury or death associated with such unint ended or unauthorised \* use, even if such claim alleges that Motorola was negligent \* regarding the design or manufacture of the part. Motorola and the \* Motorola logo\* are registered trademarks of Motorola Ltd. 

#### THERMO.ASM

\*\*\*\*\*\*\* SYSTEM EOUATES \*\*\*\*\*\*\*\*

PORTA EQU	\$00	; Port A register
PORTB EQU	\$01	; Port B register
DDRA EQU	\$04	; Port A Data Direction register
DDRB EQU	\$05	; Port B Data Direction register
ERROR EQU	0	; Error Bit
DQ EQU	5	; 1820 DQ signal
DQ_CTRL EQU	5	
SKIPROM EQU	\$CC	; 1820 Skip ROM command byte
CONVERT EQU	\$44	; 1820 Temperature Convert command byte
READRAM EQU	\$BE	; 1820 Read RAM command byte
CE EQU	\$02	; ISD2560 chip enable bit
PD EQU	\$03	; ISD2560 powerdown bit
EOM EQU	\$04	; ISD2560 end of message bit
DDRAMSK EQU	\$FF	; Port A Data Direction register mask
DDRBMSK EQU	\$2F	; Port B Data Direction register mask
PORTAMSK EQU	\$00	; Port A mask
PORTBMSK EQU	\$2C	; Port B mask
POSITIVE_SIGN	EQU	\$00 ; MSB of a positive temperature reading
NEGATIVE_SIGN	EQU	\$FF ; MSB of a neagtive temperature reading

POSITIVE_LIMIT NEGATIVE_LIMIT	EQU EQU	\$FA \$92	; The highest LSB for a positive temperature. ; The lowest LSB for a negative temperature.
******* VARIA	BLES **	* * * * * * *	
	ORG \$C	0	
SYS_STATUS TEMP_HI TEMP_LO TEMP TEMPA TEMPX	DS DS DS DS DS DS	1 1 1 1 1	<pre>; System status variable ; Stores the temperature reading high byte ; Stores the temperature reading low byte ; Temporary storage space ; Register A tempoary storage space ; Register X temporary storage space</pre>
RAW_TEMP PHRASE_BUFFER POINT_FLAG QUOTIENT PHRASE_POINTER	EQU DS DS DS DS ORG \$3	TEMP_HI \$11 1 1 1 00	; Storage space for converted reading ; Stores addresses of phrases to be output ; Flag indicating a .5 increment in temperature ; Storage space for the result of division ; Pointer to current address in phrase buffer
START: WAIT4INT	JSR STOP BRA	INITIA WAIT4I	;Stop
IRQ_INT:	CLR JSR BRSET JSR BRSET JSR	GET_TE	CE ; Debounce the activation switch SYS_STATUS,IRQ_INT_EXIT ; If the error bit is ; set, the exit routine EMP ; Get a temperature reading from the 1820 SYS_STATUS,IRQ_INT_EXIT ; If the error bit is ; set, the exit routine
IRQ_INT_EXIT	JSR BCLR RTI	OUTPUT_' ERROR,S	TEMP ; Audibly output temperature YS_STATUS ; Clear the error bit

* * * * * * * * * * * * *	* * * * * * *	* * * * * * * * * * * * *	* * * * * * * * * * * * * * * * * * * *					
*			r t					
* Function Na	Function Name: OUTPUT_TEMP *							
* Function In	Function Inputs: None *							
* Functions (	Outputs	s: None	ډ					
*			,					
* Purpose: Th	nis fun	ction output	s the contents of the *					
* phrase_buf:	fer to	the ISD2560	which outputs them *					
* audibly.			ډ					
*			*					
* * * * * * * * * * * * *	* * * * * * *	* * * * * * * * * * * * *	* * * * * * * * * * * * * * * * * * * *					
OUTPUT_TEMP:		•	; Take the ISD2560 out of powerdown mode.					
	LDX		TER ; Point to the phrase buffer.					
OUT_PHRASE:		PORTB						
		#\$FC						
	-	, X						
	STA	PORTB						
	INCX							
	LDA	•	; Put the address of the next phrase to					
	STA INCX	PORTA	; be output on the address bus of the ISD2560					
		CE, PORTB	; Pulse the ISD2560's chip enable pin to start					
	BSET	CE, PORTB	; outputting the current phrase.					
EOM H WAIT:			DM_H_WAIT ; Wait for the ISD2560's End of Message					
EOM L WAIT:			DM_L_WAIT ; pulse before continuing					
HOW_H_WATI	LDA	.Х	; Look for the end of the phrases to be output					
	CMP	#\$FF	; if it is found exit the routine. Otherwise					
	BNE	OUT PHRASE						
	BNE	PD, PORTB	; Put the ISD2560 into powerdown mode.					
	RTS		, rae ene ibbzood into powerdown mode.					
	1110							

* * * * * * * * * * * *	* * * * * * *	* * * * * * * * * * * * * *	***************************************
*			*
* Function N	Name: F	ORM_PHRASE	*
* Function 1	Inputs:	None	*
* Functions	Output	s: None	*
*			*
* Purpose: 1	This fu	nction conver	ts the temperature read *
* from the 1	1820 to	the addresse	s of the phrases in *
* the ISD256	50 that	match the ind	dividual digits in the *
* reading. T	These a	ddresses are	stored in the phrase *
* buffer.			*
*			*
********	* * * * * * *	* * * * * * * * * * * * * * *	***************************************
FORM_PHRASE:	CLR	POINT_FLAG	; Check to see if the temperature reading is a
			; a .5 increment, if it is set the POINT_FLAG.
		R 0,(RAW_TEMP+	-1),NOT_POINT
	INC	POINT_FLAG	
NOT_POINT:	LDX	#PHRASE_BUFF	
	LDA	RAW_TEMP	; Check to see if the temperature is negative
	BEQ	NOT_NEG	
Otherwise	LDA	NEG_ADDR	; phrase at the start of the phrase buffer.
Otherwise	CULY	v	· convert the temperature into its positive
	STA	, X	; convert the temperature into its positive equivalent.
	INCX		
	LDA	(NEG ADDR+1)	
	STA	, X	
	INCX		
	COM	(RAW_TEMP+1)	)
	INC	(RAW_TEMP+1)	)
NOT_NEG:	LSR	(RAW_TEMP+1)	; Check for the temperature being lower than 100
degrees			
	LDA	(RAW_TEMP+1)	; Celcius.
	CMP	#\$64	
	BLO	BELOW_100	
	SUB	#\$64	
	STA	(RAW_TEMP+1)	i de la construcción de la constru
	LDA	HUNDRED_ADD	R ; If the temperature is greater than or equal to
			; 100 degrees
	STA	, X	; put the address of the "One hundred" phrase in
			the phrase
	INCX		; buffer and subtract the equivalent value of 100
			from the value.
	LDA	(HUNDRED_ADDF	(+⊥)
	STA	, X	
	INCX	( האודות הוארה . 1 \	
	LDA	(RAW_TEMP+1)	
	BEQ	POINT	

BELOW_100:	LDA	(RAW_TEMP+1) ;	Check to see if the remaining temperature value is less than 20
	CMP	#\$14	; degrees. If it is, search for it in the TB0_19 table.
	BLO	BELOW_20	; Otherwise divide the data by ten. Store the
	CLR	QUOTIENT	<pre>quotient in the ; quotient variable and the remainder in  (RAW_TEMP+1).</pre>
	SUB	#\$14	(RAW_1EMP+1).
DIV10	CMP	#\$1 #\$A	
21110	BLO	DIV_DONE	
	INC	QUOTIENT	
	SUB	~ #\$A	
	BRA	DIV10	
DIV_DONE	STA	(RAW_TEMP+1)	
	ASL	QUOTIENT	
	STX	PHRASE_POINTER	R ; Find the address of the quotient's phrase in
	LDX	QUOTIENT	;the TBL20_90 table and store it in the phrase
buffer.			
	LDA	TBL20_90,X	
	INCX		
	STX	TEMP	
	LDX	PHRASE_POINTER	ł
	STA	, X	
	INCX		
	STX LDX	PHRASE_POINTER TEMP	ζ.
	LDA	TBL20_90,X	
	LDX	PHRASE_POINTER	
	STA	, X	<b>`</b>
	INCX		
	LDA	(RAW_TEMP+1)	
	BEQ	POINT	
BELOW_20	LDA	(RAW_TEMP+1) ; the	Find the address of the remainder's phrase in
	ASLA		TBL0_19 table and store it in the phrase
			buffer.
	STX	PHRASE_POINTER	
	TAX		
	LDA	TBL0_19,X	
	INCX		
	STX	TEMP	
	LDX	PHRASE_POINTER	R
	STA	, X	
	INCX		
	STX	PHRASE_POINTER	
	LDX	TEMP TRIO 19 Y	
	LDA LDX	TBL0_19,X PHRASE_POINTER	
	STA	,X	
	INCX		

TST POINT\_FLAG POINT ; If the temperature is a .5 increment reading BEQ END RAWTEMP ; load the phrase buffer with the addresses for the LDA POINT\_ADDR ; "Point" and "Five" phrases. STA ,Х INCX (POINT\_ADDR+1) LDA STA ,Х INCX LDA FIVE\_ADDR STA ,Х INCX LDA (FIVE ADDR+1) STA , X INCX END\_RAWTEMP LDA DEGREE\_ADDR ; Load the phrase buffer with the address for STA , X ; the "Degrees" phrase. INCX LDA (DEGREE\_ADDR+1) STA , X INCX LDA CELCIUS\_ADDR ; Load the phrase buffer with the address for STA ; the "Celcius" phrase. , X INCX LDA (CELCIUS\_ADDR+1) STA , X INCX CLR ,Х DEC , X RTS \* \* \* \* Function Name: INITIALIZE \* \* Function Inputs: None \* Functions Outputs: None \* \* \* Purpose: This function configures PORT A and PORT B \* \* \* and their data direction registers. \* INITIALIZE #DDRAMSK LDA STA DDRA LDA #PORTAMSK STA PORTA LDA #DDRBMSK STA DDRB LDA **#PORTBMSK** PORTB STA RTS

* * * * * * * * * * * * *	* * * * * * * * *	* * * * * * * * * * * * * * * * * *	* * * * * * * * * * * * * * * * * * * *							
*			*							
* Function Na	ame: GET I	EMP	*							
* Function In			*							
	Functions Outputs: None									
*	oucpueb i	one	×							
* Purpose: T	his functi	on performs the r	required reads and *							
			erature conversion *							
		e temperature read								
* TEMP varia		comperature read	* 15 100011100 111							
*	bic.		*							
* * * * * * * * * * * * *	* * * * * * * * * *	* * * * * * * * * * * * * * * * * *	*****							
GET TEMP	JSR	RESET_1820	; Reset the 1820.							
	BRSET	ERROR, SYS STATU								
	LDA	· _	; Send the 1820's SKIP ROM command.							
	STA	TEMP	/ Send the 1820 S Skip Kom command.							
	JSR	WRITE_1820	· Cond the 102015 CONTERPORT sommand							
	LDA		; Send the 1820's CONVERT T command.							
	STA	TEMP								
	JSR	WRITE_1820								
READ_LOOP	JSR	READ_1820								
	LDA	TEMP								
	CMP	#\$FF								
	BNE	READ_LOOP								
	JSR	RESET_1820	; Reset the 1820.							
	BRSET	ERROR, SYS_STA	TUS,GET_ERROR ; If the reset fails set the							
	LDA	#SKIPROM	; error bit and exit the routine.							
	STA	TEMP	; Send the 1820's SKIP ROM command.							
	JSR	WRITE_1820								
	LDA	#READRAM	vcccccccc; Read the 1820's RAM to get the temperature							
	STA	TEMP	; reading.							
	JSR	WRITE_1820	/ Teading.							
	JSR	READ_1820								
	LDA	TEMP								
	STA	TEMP_LO								
	JSR	READ_1820								
	LDA	TEMP								
	STA	TEMP_HI								
	CMP	#POSITIVE_SIGN	; Check for an invalid positive							
	BEQ	CHK_POSITIVE	; data value.							
	CMP	#NEGATIVE_SIGN	; Check for an invalid negative							
	BNE	GET_ERROR	; data value.							
	LDA	TEMP_LO								
	CMP	#NEGATIVE_LIMIT								
	BLO	GET_ERROR								
	BRA	GET_EXIT								

```
CHK_POSITIVE LDA
                  TEMP_LO
             CMP
                   #POSITIVE LIMIT
             BLS
                  GET EXIT
GET ERROR
             BSET ERROR, SYS_STATUS ; Set the error bit if an error
                   RESET_1820 ; occurs.
GET_EXIT
             JSR
             RTS
*
* Function Name: RESET_1820
                                                                    *
* Function Inputs: None
                                                                    *
* Functions Outputs: None
                                                                    *
* Purpose: This function resets the 1820. If the 1820
                                                                    *
* resets properly, it will return a response pulse. If
                                                                    *
* a pulse is not received, the error bit is set in
* system status.
                                                                    *
RESET_1820
             STA
                  TEMPA
                                 ; Save the CPU registers
             STX
                   TEMPX
             BSET
                   DQ,PORTB
                                ; Send a reset pulse to
                   DQ CTRL,DDRB
             BSET
                                 ; the 1820
             BCLR DQ, PORTB
                   DELAY 500uS
             JSR
             BSET
                   DQ,PORTB
                   DQ_CTRL,DDRB ; Set the J1A to receive the
DELAY_100uS ; response pulse from the 1820
             BCLR
             JSR
             BRSET DQ, PORTB, RESET_ERR ; If the start of the pulse
                                ; is not received, handle the error
             JSR
                   DELAY 500uS
             BRSET DQ, PORTB, RESET_EXIT
RESET ERR
             BSET ERROR, SYS_STATUS ; Set the error bit
RESET_EXIT
                   DQ,PORTB ; Set the J1A for transmission
             BSET
             BSET
                   DQ CTRL,DDRB
             LDA
                   TEMPA
                                 ; Restore CPU registers
                   TEMPX
             LDX
             RTS
```

#### Source Code

\* \* \* Function Name: WRITE 1820 \* \* Function Inputs: None \* \* Functions Outputs: None \* Purpose: This function writes the data stored in the \* \* TEMP variable to the 1820. WRITE 1820 STA TEMPA ; Save CPU registers. STX TEMPX LDX #8 ; Load X with count. ; Shift out the bit to be sent WRITE SHIFT LSR TEMP BCS WRITE ONE BCLR DQ,PORTB JSR DELAY\_80uS WRITE\_ZERO ; Send a zero to the 1820 BSET DQ, PORTB DEC WRITE BRA BCLR DQ, PORTB ; Send a one to the 1820 WRITE ONE NOP NOP NOP BSET DQ,PORTB JSR DELAY 80uS DEC WRITE DECX WRITE\_SHIFT BNE LDA TEMPA ; Restore CPU registers TEMPX LDX RTS \* \* \* Function Name: READ\_1820 \* Function Inputs: None \* \* Functions Outputs: None \* \* \* Purpose: This function reads data from the 1820 and returns the data in the TEMP variable. \* READ\_1820 STA TEMPA ; Save CPU registers STX TEMPX LDX #8 ; Load X registers with count

READ_BIT	BSET BSET BCLR NOP NOP NOP NOP NOP	DQ,PORTB ; DQ_CTRL,DDRB DQ,PORTB	; Set up the DQ line for read
	BCLR BRSET CLC BRA	DQ_CTRL,DDRB ; DQ,PORTB,READ_ONE READ_SHIFT	; Set the DQ line to receive data NE ; Read bit
READ_ONE READ_SHIFT	SEC ROR JSR	TEMP ; DELAY_80uS	; Rotate the bit in the TEMP variable
	DECX BNE BSET BSET LDA LDX RTS	READ_BIT DQ,PORTB DQ_CTRL,DDRB TEMPA ; TEMPX	; Restore CPU registers
<pre>************************************</pre>	e: DEBOU	NCE e	**************************************
*			<pre>pushbutton switch. * * *********************************</pre>
DEBOUNCE DEBOUNCE_LOOP	LDX JSR DECX BNE BIL	#\$FF DELAY_500uS DEBOUNCE_LOOP DEBOUNCE_EXIT	; If the interrupt is valid, exit
	BSET		; the routine S ; If the interrupt is invalid, set ; the error bit and exit
DEBOUNCE_EXIT	RTS		

	* * * * * *	* * * * * *	* * * * * * * * * *	* * * * * * * * *	* * * * *	* * * * *	* * * * * * * *	*******	**
*	<b>.</b>		_						*
* Function 1									*
* Functions *	Outpu	ICS: NO	one						*
									*
* Purpose: 7	Inis I	unctio	on provides	delays.					*
								* * * * * * * * * * * * * * * * * * * *	*
* * * * * * * * * * * * * *	****	* * * * * *	* * * * * * * * * * *	* * * * * * * * *			* * * * * * * * *	* * * * * * * * * * * * * * * * * * * *	* *
DELAY_80uS	LI	DA	#\$0C						
	BI	RA	DELAY_LOO	P					
DELAY_100uS	$\mathbf{LI}$	DA	#\$0F						
	BI	RA	DELAY_LOO	P					
DELAY_500uS	$\mathbf{LI}$	DA	#\$52						
	BI	RA	DELAY_LOO	P					
DELAY_LOOP	NC	ЭР							
	NC	ЭР							
	NC	ЭР							
	DI	ECA							
	BI	NE	DELAY_LOO	P					
	R	ΓS							
* * * * * * * * * * * * * * *	* * * * * *	* * * * * *	* * * * * * * * * *	* * * * * * * * *	* * * * *	* * * * *	* * * * * * * *	* * * * * * * * * * * * * * * * * * * *	**
*	P	HRASE	ADDRESS TA	BLE					*
*	-		1001100 11						*
* * * * * * * * * * * *	* * * * * *	* * * * * *	* * * * * * * * * *	* * * * * * * * *	* * * * *	****	* * * * * * * *	* * * * * * * * * * * * * * * * * * * *	* *
	org	\$700							
TBL0 19:	DH	40000		Jalanaaa	£	+ h a		" <b>7</b> and "	
1800_19.	DW DW	\$0000 \$0010		Address Address					
	DW DW	\$0010 \$0020		Address			-		
	DW DW	\$0020						"Three".	
	DW DW	\$0030		Address					
FIVE ADDR:	DW DW	\$0040		Address					
FIVE_ADDR.	DW	\$0050		Address			-		
	DW	\$0070						"Seven".	
	DW	\$0080					-	"Eight".	
	DW	\$0090		Address					
	DW	\$00A0		Address			-		
	DW	\$00B0					-	"Eleven".	
	DW	\$00C0						"Twelve".	
	DW	\$00C0 \$00D0						"Thirteen".	
	DW	\$00E0						"Fourteen".	
	DW	\$00E0					-	"Fifteen".	
	DW	\$0100						"Sixteen".	
	DW	\$0110	;	Address	for	the	phrase	"Seventeen".	

	DW	\$0120	;	Address	for	the	phrase	"Eighteen".
	DW	\$0130	;	Address	for	the	phrase	"Nineteen".
TBL20_90:	DW	\$0140	;	Address	for	the	phrase	"Twenty".
	DW	\$0150	;	Address	for	the	phrase	"Thirty".
	DW	\$0160	;	Address	for	the	phrase	"Forty".
	DW	\$0170	;	Address	for	the	phrase	"Fifty".
	DW	\$0180	;	Address	for	the	phrase	"Sixty".
	DW	\$0190	;	Address	for	the	phrase	"Seventy".
	DW	\$01A0	;	Address	for	the	phrase	"Eighty".
	DW	\$01B0	;	Address	for	the	phrase	"Ninety".
HUNDRED_ADDR:	DW	\$01C0	;	Address	for	the	phrase	"One Hundred".
POINT_ADDR:	DW	\$01D0	;	Address	for	the	phrase	"Point".
DEGREE_ADDR:	DW	\$01E0	;	Address	for	the	phrase	"Degree".
NEG_ADDR:	DW	\$01F0	;	Address	for	the	phrase	"Negative".
CELCIUS_ADDR:	DW	\$0200	;	Address	for	the	phrase	"Celcius".

ORG	\$7FA
DW	IRQ_INT
ORG	\$7FE
DW	START

END

Motorola reserves the right to make changes without further notice to any products herein. Motorola makes no warranty, representation or guarantee regarding the suitability of its products for any particular purpose, nor does Motorola assume any liability arising out of the application or use of any product or circuit, and specifically disclaims any and all liability, including without limitation consequential or incidental damages. "Typical" parameters which may be provided in Motorola data sheets and/or specifications can and do vary in different applications and actual performance may vary over time. All operating parameters, including "Typicals" must be validated for each customer application by customer's technical experts. Motorola does not convey any license under its patent rights or the rights of others. Motorola products are not designed, intended, or authorized for use as components in systems intended for surgical implant into the body, or other applications intended to support or sustain life, or for any other application in which the failure of the Motorola product could create a situation where personal injury or death may occur. Should Buyer purchase or use Motorola products for any such unintended or unauthorized application, Buyer shall indemnify and hold Motorola and its officers, employees, subsidiaries, affiliates, and distributors harmless against all claims, costs, damages, and expenses, and reasonable attorney fees arising out of, directly or indirectly, any claim of personal injury or death massociated with such unintended or unauthorized use, even if such claim alleges that Motorola was negligent regarding the design or manufacture of the part. Motorola and (A) are registered trademarks of Motorola, Inc. Motorola, Inc. is an Equal Opportunity/Affirmative Action Employer.

#### How to reach us:

USA/EUROPE/Locations Not Listed: Motorola Literature Distribution; P.O. Box 20912; Phoenix, Arizona 85036.1-800-441-2447 or 602-303-5454

MFAX: RMFAX0@email.sps.mot.com – TOUCHTONE 602-244-6609

INTERNET: http://Design-NET.com

JAPAN: Nippon Motorola Ltd.; Tatsumi-SPD-JLDC, 6F Seibu-Butsuryu-Center, 3-14-2 Tatsumi Koto-Ku, Tokyo 135, Japan. 03-81-3521-8315

ASIA/PACIFIC: Motorola Semiconductors H.K. Ltd.; 8B Tai Ping Industrial Park, 51 Ting Kok Road, Tai Po, N.T., Hong Kong. 852-26629298

